Title stata.com

eteffects — Endogenous treatment-effects estimation

Description Quick start Menu Syntax

Options Remarks and examples Stored results Methods and formulas

Acknowledgment References Also see

# Description

eteffects estimates the average treatment effect (ATE), the average treatment effect on the treated (ATET), and the potential-outcome means (POMs) from observational data when treatment assignment is correlated with the potential outcomes. It allows for continuous, binary, count, fractional, and nonnegative outcomes and requires a binary treatment. To control for the endogeneity of the treatment assignment, the estimator includes residuals from the treatment model in the models for the potential outcomes, known as a control-function approach.

## **Quick start**

```
ATE of binary treatment treat using a linear model for outcome y1 on x and the residuals from a probit model for treat on x and z
```

```
eteffects (y1 x) (treat x z)
```

Same as above, but estimate ATET

```
eteffects (y1 x) (treat x z), atet
```

Same as above, but estimate POMs

```
eteffects (y1 x) (treat x z), pomeans
```

Same as above, and show parameters from auxiliary equations

```
eteffects (y1 x) (treat x z), pomeans aequations
```

ATE of treat using an exponential-mean model for y1

```
eteffects (y1 x, exponential) (treat x z)
```

Same as above, but for count outcome y2

```
eteffects (y2 x, exponential) (treat x z)
```

Same as above, but use a probit model for binary outcome y3

```
eteffects (y3 x, probit) (treat x z)
```

Same as above, but use a fractional probit model for y4 ranging from 0 to 1

```
eteffects (y4 x, fractional) (treat x z)
```

#### Menu

 ${\it Statistics} > {\it Causal inference/treatment effects} > {\it Endogenous treatment} > {\it Control function estimator} > {\it Continuous outcomes}$ 

 ${\tt Statistics} > {\tt Causal inference/treatment effects} > {\tt Endogenous treatment} > {\tt Control function estimator} > {\tt Binary outcomes}$ 

 ${\tt Statistics} > {\tt Causal \ inference/treatment \ effects} > {\tt Endogenous \ treatment} > {\tt Control \ function \ estimator} > {\tt Count \ outcomes}$ 

Statistics > Causal inference/treatment effects > Endogenous treatment > Control function estimator > Fractional outcomes

Statistics > Causal inference/treatment effects > Endogenous treatment > Control function estimator > Nonnegative outcomes

# **Syntax**

```
eteffects (ovar omvarlist [, omodel noconstant])
  (tvar tmvarlist [, noconstant]) [if] [in] [weight] [, stat options]
```

ovar is the depvar of the outcome model.

omvarlist is the list of exogenous indepvars in the outcome model.

tvar is the binary treatment variable.

tmvarlist is the list of covariates that predict treatment assignment.

omodel	Description				
Model					
linear	linear outcome model; the default				
fractional	fractional probit outcome model				
probit	probit outcome model				
exponential	exponential-mean outcome model				
stat Description					
Model					
ate	estimate average treatment effect in population; the default				
atet	estimate average treatment effect on the treated				
$\underline{\mathtt{pom}}\mathtt{eans}$	estimate potential-outcome means				

options	Description					
Model						
<u>nocons</u> tant	suppress constant term					
SE/Robust						
vce(vcetype)	<pre>vcetype may be robust, cluster clustvar, bootstrap, or jackknife</pre>					
Reporting						
<u>l</u> evel(#)	set confidence level; default is level(95)					
aequations	display auxiliary-equation results					
display_options	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling					
Maximization						
maximize_options	control the maximization process; seldom used					
Advanced						
<pre>pstolerance(#)</pre>	set tolerance for overlap assumption					
$\frac{\overline{\text{os}}}{\text{os}}$ ample(newvar)	generate newvar to mark observations that violate the overlap assumption					
<u>coefl</u> egend	display legend instead of statistics					

omvarlist and tmvarlist may contain factor variables; see [U] 11.4.3 Factor variables.

bootstrap, by, collect, jackknife, and statsby are allowed; see [U] 11.1.10 Prefix commands.

Weights are not allowed with the bootstrap prefix; see [R] bootstrap.

fweights, iweights, and pweights are allowed; see [U] 11.1.6 weight.

coeflegend does not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

# **Options**

```
noconstant; see [R] Estimation options.
```

noconstant, see [K] Estimation options.

stat is one of three statistics: ate, atet, or pomeans. ate is the default.

ate specifies that the average treatment effect be estimated.

atet specifies that the average treatment effect on the treated be estimated.

pomeans specifies that the potential-outcome means for each treatment level be estimated.

```
SE/Robust
```

vce(*vcetype*) specifies the type of standard error reported, which includes types that are robust to some kinds of misspecification (robust), that allow for intragroup correlation (cluster *clustvar*), and that use bootstrap or jackknife methods (bootstrap, jackknife); see [R] *vce\_option*.

```
Reporting
```

level(#); see [R] Estimation options.

aequations specifies that the results for the outcome-model or the treatment-model parameters be displayed. By default, the results for these auxiliary parameters are not displayed.

display\_options: noci, nopvalues, noomitted, vsquish, noemptycells, baselevels, allbaselevels, nofvlabel, fvwrap(#), fvwrapon(style), cformat(%fmt), pformat(%fmt), sformat(%fmt), and nolstretch; see [R] Estimation options.

```
Maximization

maximize_options: iterate(#), [no]log, and from(init_specs); see [R] Maximize. These options
are seldom used.
init_specs is one of
    matname [, skip copy]

# [, # ...], copy
```

pstolerance(#) specifies the tolerance used to check the overlap assumption. The default value is pstolerance(1e-5). eteffects will exit with an error if an observation has an estimated propensity score smaller than that specified by pstolerance().

osample(newvar) specifies that indicator variable newvar be created to identify observations that violate the overlap assumption.

The following option is available with eteffects but is not shown in the dialog box:

coeflegend; see [R] Estimation options.

# Remarks and examples

Advanced

stata.com

If you are unfamiliar with treatment-effects estimators for observational data or the teffects commands, we recommend that you look at [CAUSAL] teffects intro. For the intuition behind some of the concepts discussed below, we recommend that you read *Defining treatment effects* in [CAUSAL] teffects intro advanced.

The estimators implemented in eteffects extend the regression adjustment (RA) estimators implemented in teffects ra to allow for endogenous treatments, that is, when treatment assignment is not independent of outcomes. This endogeneity is a violation of the conditional mean independence assumption used by teffects ra, as discussed in *The potential-outcome model* in [CAUSAL] teffects intro advanced.

eteffects estimates the average treatment effect (ATE), the average treatment effect on the treated (ATET), and the potential-outcome means (POMs). It uses a linear, a probit, a fractional probit, or an exponential-mean model for the potential outcomes and a probit model for treatment assignment. After conditioning on the observable covariates, eteffects allows some remaining unobservable components to affect both treatment assignment and the potential outcomes. The treatment assignment process is endogenous because these unobservable components affect both treatment assignment and the potential outcomes.

To control for the endogeneity of the treatment assignment, eteffects uses a control-function approach. This method controls for endogeneity by including the residuals from the treatment-assignment model as a regressor in the models for the potential outcome. The implementation in eteffects follows Wooldridge (2010), who provides an excellent discussion of the control-function approach that addresses endogeneity problems in a treatment-effects context.

The control-function approach estimates the parameters of the conditional means of the potential outcomes. Sample averages of the conditional means are used to estimate the unconditional ATE, ATET, or POMs. This method is known as RA.

Taken collectively, the estimators implemented in eteffects are control-function RA estimators. See *Methods and formulas* below for details about the estimation procedure.

#### Example 1: Linear outcome estimates for ATE

Nonsmoker

3437.964

31.21145

Suppose we want to know the effect of a mother smoking while pregnant on the birthweight of her infant. We use an extract from Cattaneo (2010) in which bweight records the baby's birthweight and mbsmoke is the variable (0 or 1) indicating whether a mother smoked while pregnant.

We may believe that birthweight (the potential outcome) is influenced by whether the mother had a prenatal exam in the first trimester, whether the mother is married, the mother's age, whether this is the first birth, and the education level of the father. We may also believe that the smoking decision (the treatment) is influenced by the mother's marital status, the education level of the mother, her age, whether she had a prenatal exam in the first trimester, and whether this baby is her first baby.

Thus we condition on different sets of covariates in the models for treatment assignment and the potential outcomes. In the probit model for smoking status (mbsmoke), we condition on marital status (mmarried), age (mage), mother's education level (medu), father's education level (fedu), and whether it was the mother's first baby (fbaby). We model birthweight (bweight) as a linear function of whether the mother had a first-trimester prenatal exam (prenatal1), mmarried, mage, and fbaby. We can estimate the ATE of smoking status using one of the teffects estimators if we believe that there are no unobservable components that affect both the decision to smoke while pregnant and the potential birthweights.

If we believe there is some unobservable factor that affects both assignment to treatment and the potential outcome, we must select another estimator. For example, we do not observe a mother's health consciousness, which affects both the smoking decision and each potential birthweight through other behaviors such as intake of prenatal vitamins. Under these assumptions, the estimators in eteffects consistently estimate the ATE, but the estimators in [CAUSAL] teffects yield inconsistent estimates.

```
. use https://www.stata-press.com/data/r18/cattaneo2
(Excerpt from Cattaneo (2010) Journal of Econometrics 155: 138-154)
. eteffects (bweight i.prenatal1 i.mmarried mage i.fbaby)
> (mbsmoke i.mmarried mage i.fbaby medu fedu)
Iteration 0: EE criterion = 4.704e-24
Iteration 1: EE criterion = 1.223e-25
Endogenous treatment-effects estimation
                                                          Number of obs = 4,642
Outcome model: linear
Treatment model: probit
                              Robust
                                                 P>|z|
                                                            [95% conf. interval]
     bweight
               Coefficient
                            std. err.
                                            z
ATF.
     mbsmoke
    (Smoker
         779
                -455.9119
                            212.4393
                                                          -872.2853
Nonsmoker)
                                         -2.15
                                                 0.032
                                                                       -39.53852
POmean
     mbsmoke
```

When no mother smokes, the average birthweight is 3,438 grams. The average birthweight is 456 grams less when all mothers smoke than when no mother smokes.

110.15

0.000

3376.791

3499.138

We can compare these results with those obtained if we ignore the endogeneity of the smoking decision. Below we estimate the ATE using the inverse-probability-weighted regression-adjustment estimator in [CAUSAL] teffects ipwra.

. teffects ipwra (bweight i.prenatal1 i.mmarried mage i.fbaby)

> (mbsmoke i.mmarried mage i.fbaby medu fedu)

Iteration 0: EE criterion = 3.036e-22 Iteration 1: EE criterion = 3.755e-26

Treatment-effects estimation Number of obs 4,642

: IPW regression adjustment Estimator

Outcome model : linear Treatment model: logit

bweight	Coefficient	Robust std. err.	z	P> z	[95% conf.	interval]
ATE mbsmoke (Smoker vs Nonsmoker)	-233.6835	25.07695	-9.32	0.000	-282.8335	-184.5336
POmean mbsmoke Nonsmoker	3403.191	9.529709	357.11	0.000	3384.513	3421.869

In magnitude, the estimated ATE is more than half the estimate that allows for endogenous treatment assignment. If there is endogeneity, disregarding it underestimates the effect of smoking on birthweight, We show how to test for endogeneity in example 1 of [CAUSAL] eteffects postestimation.

# Example 2: Estimating the ATET

Continuing example 1, we can use the atet option to estimate the ATET.

. eteffects (bweight i.prenatal1 i.mmarried mage i.fbaby)

> (mbsmoke i.mmarried mage i.fbaby medu fedu), atet

Iteration 0: EE criterion = 4.688e-24 Iteration 1: EE criterion = 8.479e-26

Endogenous treatment-effects estimation Number of obs = 4,642

Outcome model: linear Treatment model: probit

bweight	Coefficient	Robust std. err.	z	P> z	[95% conf.	interval]
ATET  mbsmoke (Smoker  vs Nonsmoker)	-409.8527	161.4816	-2.54	0.011	-726.3507	-93.35466
POmean mbsmoke Nonsmoker	3547.512	160.0595	22.16	0.000	3233.801	3861.223

4

4

In the population of mothers who smoke, the average infant birthweight would be 3,548 grams if none of these mothers smoked. For the mothers who smoke, the average infant birthweight is 410 grams less than if none of these mothers smoked.

## Example 3: Exponential-mean outcomes

We estimate the ATE of living in an urban area on monthly earnings (wage), using a subset of the National Longitudinal Survey in 1980 found in Wooldridge (2010). We assume that once we condition on work experience (exper), whether education level attained is college or higher (college), and IQ (iq), individual wages follow an exponential mean. The variables used to predict residence in an urban area (urban) are college and whether the respondent's father attained a bachelor's degree or higher (fcollege).

wage	Coefficient	Robust std. err.	z	P> z	[95% conf.	interval]
ATE urban (1 vs 0)	481.0465	31.74882	15.15	0.000	418.82	543.2731
POmean urban 0	233.8083	13.51028	17.31	0.000	207.3286	260.288

When everyone lives outside urban areas, wages are \$234 a month on average. Wages are \$481 a month greater, on average, when everyone lives in urban areas.

1

## Stored results

eteffects stores the following in e():

```
Scalars
    e(N)
                             number of observations
    e(ni)
                             number of observations for treatment level j
                             number of equations in e(b)
    e(k_eq)
    e(k_levels)
                             number of levels in treatment variable
    e(rank)
                             rank of e(V)
    e(converged)
                             1 if converged, 0 otherwise
Macros
    e(cmd)
                             eteffects
    e(cmdline)
                             command as typed
    e(depvar)
                             name of outcome variable
                             name of treatment variable
    e(tvar)
```

```
8
```

```
e(omodel)
                             fractional, linear, probit, or exponential
    e(stat)
                             statistic estimated, ate, atet, or pomeans
    e(wtype)
                             weight type
    e(wexp)
                             weight expression
    e(title)
                             title in estimation output
    e(tlevels)
                             levels of treatment variable
    e(vce)
                             vcetype specified in vce()
    e(vcetype)
                             title used to label Std. err.
    e(properties)
    e(estat_cmd)
                             program used to implement estat
    e(predict)
                             program used to implement predict
    e(marginsnotok)
                             predictions disallowed by margins
    e(asbalanced)
                             factor variables fvset as asbalanced
    e(asobserved)
                             factor variables fyset as asobserved
Matrices
    e(b)
                             coefficient vector
    e(V)
                             variance-covariance matrix of the estimators
Functions
    e(sample)
                             marks estimation sample
```

In addition to the above, the following is stored in r():

```
Matrices
r(table) matrix containing the coefficients with their standard errors, test statistics, p-values,
and confidence intervals
```

Note that results stored in r() are updated when the command is replayed and will be replaced when any r-class command is run after the estimation command.

## Methods and formulas

The treatment-effects models considered in eteffects are given by

$$y_{i0} = E\left(y_{i0}|\mathbf{x}_i\right) + \epsilon_{i0} \tag{1}$$

$$y_{i1} = E\left(y_{i1}|\mathbf{x}_i\right) + \epsilon_{i1} \tag{2}$$

$$t_i = E\left(t_i|\mathbf{z}_i\right) + \nu_i \tag{3}$$

$$y_i = t_i y_{i1} + (1 - t_i) y_{i0} (4)$$

$$E\left(\epsilon_{ij}|\mathbf{x}_{i},\mathbf{z}_{i}\right) = E\left(\epsilon_{ij}|\mathbf{z}_{i}\right) = E\left(\epsilon_{ij}|\mathbf{x}_{i}\right) = 0 \quad \text{for} \quad j \in \{0,1\}$$
(5)

$$E\left(\epsilon_{ij}|t\right) \neq 0 \quad \text{for} \quad j \in \{0,1\} \tag{6}$$

where the subscript i denotes individual level observations,  $y_{i1}$  is the potential outcome of receiving the treatment,  $y_{i0}$  is the potential outcome when the treatment is not received,  $t_i$  is the observed binary treatment, and  $y_i$  is the observed outcome. Each one of the potential outcomes is determined by its expected value conditional on a set of regressors  $\mathbf{x}_i$  and an unobserved random component  $\epsilon_{ij}$ , for  $j \in \{0,1\}$ . Similarly, the treatment is given by its expectation conditional on a set of regressors  $\mathbf{z}_i$ , which does not need to differ from  $\mathbf{x}_i$ , and an unobserved component  $\nu_i$ .

Equations (1)–(5) describe the parametric treatment-effects models in [CAUSAL] **teffects**. Equation (6) adds endogeneity to the framework. It states that the unobservables in the potential-outcome equations are correlated to treatment status. For our birthweight example, this would happen if mothers who do not smoke are more health conscious than those who smoke and if we do not observe health awareness in our data. If we do not observe health awareness, the decision to smoke or not to smoke is not independent of the infant's birthweight.

Equations (3), (5), and (6) are the basis of the control-function estimator implemented by eteffects. Equation (5) states that the unobserved components in the potential outcome are independent of  $\mathbf{z}_i$ . Therefore, the correlation between  $t_i$  and the unobserved components must be equivalent to the correlation between  $\epsilon_{ij}$  and  $\nu_i$ . Another way of stating this is

from (3) 
$$E(\epsilon_{ij}|t_i) = E(\epsilon_{ij}|E(t|\mathbf{z}_i) + \nu_i)$$
 from (5) 
$$= E(\epsilon_{ij}|\nu_i)$$
 
$$= \nu_i \beta_{2i}$$

We fit (3) using a probit estimator. We then obtain  $\widehat{\nu}_i$  as the difference between the treatment and our estimate of  $E\left(t_i|\mathbf{z}_i\right)$  and use this statistic to compute an estimate of  $E\left(y_{ij}|\mathbf{x}_i,\nu_i,t_i\right)$  for  $j\in\{0,1\}$ . If the outcome is linear, for instance,

$$E(y_{ij}|\mathbf{x}_i,\nu_i,t_i=j) = \mathbf{x}_i'\boldsymbol{\beta}_{1j} + \nu_i\boldsymbol{\beta}_{2j} \quad \text{for} \quad j \in \{0,1\}$$

For the probit and exponential-mean cases, respectively, we have the following:

$$E(y_{ij}|\mathbf{x}_i,\nu_i,t_i=j) = \mathbf{\Phi}\left(\mathbf{x}_i'\boldsymbol{\beta}_{1j} + \nu_i\boldsymbol{\beta}_{2j}\right)$$
(8)

$$E(y_{ij}|\mathbf{x}_i,\nu_i,t_i=j) = \exp\left(\mathbf{x}_i'\boldsymbol{\beta}_{1j} + \nu_i\boldsymbol{\beta}_{2j}\right)$$
(9)

The parameters of (3) and (7)–(9), and the ATE, ATET, and POMs are estimated using the generalized method of moments (GMM). The moment equations used in GMM are the sample analogs of  $E\{\mathbf{w}_i'\epsilon_i(\theta)\}=0$ , where  $\mathbf{w}_i$  are the instruments,  $\epsilon_i(\theta)$  are residuals, and  $\theta$  are the parameters of the model (see [R] gmm). The moment conditions in the GMM estimation for the linear model are given by

$$\frac{1}{n}\sum_{i=1}^{n}\mathbf{x}_{i}'(y_{i}-\mathbf{x}_{i}'\widehat{\boldsymbol{\beta}}_{1j}+\widehat{\nu}_{i}\widehat{\boldsymbol{\beta}}_{2j})t_{i}=0$$
(10)

$$\frac{1}{n}\sum_{i=1}^{n}\mathbf{x}_{i}'(y_{i}-\mathbf{x}_{i}'\widehat{\boldsymbol{\beta}}_{1j}+\widehat{\nu}_{i}\widehat{\boldsymbol{\beta}}_{2j})(1-t_{i})=0$$
(11)

$$\frac{1}{n} \sum_{i=1}^{n} \mathbf{z}_{i}^{\prime} \left\{ t_{i} \frac{\boldsymbol{\phi} \left( \mathbf{z}_{i}^{\prime} \widehat{\boldsymbol{\pi}} \right)}{\boldsymbol{\Phi} \left( \mathbf{z}_{i}^{\prime} \widehat{\boldsymbol{\pi}} \right)} - (1 - t_{i}) \frac{\boldsymbol{\phi} \left( \mathbf{z}_{i}^{\prime} \widehat{\boldsymbol{\pi}} \right)}{1 - \boldsymbol{\Phi} \left( \mathbf{z}_{i}^{\prime} \widehat{\boldsymbol{\pi}} \right)} \right\} = 0$$
(12)

$$\frac{1}{n} \sum_{i=1}^{n} \left\{ \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{10} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{20} \right) - \widehat{\text{POM0}} \right\} = 0 \tag{13}$$

$$\frac{1}{n} \sum_{i=1}^{n} \left\{ \left( \mathbf{x}_{i}' \widehat{\beta}_{11} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{21} \right) - \widehat{\text{POM0}} - \widehat{\text{ATE}} \right\} = 0 \tag{14}$$

where  $\widehat{\nu}_i = t_i - \Phi\left(\mathbf{z}_i'\widehat{\pi}\right)$ , n is the number of observations, and  $\widehat{\boldsymbol{\beta}}_{11}, \widehat{\boldsymbol{\beta}}_{10}, \widehat{\boldsymbol{\beta}}_{21}, \widehat{\boldsymbol{\beta}}_{20}, \widehat{\pi}, \widehat{\text{ATE}}$ , and  $\widehat{\text{POMO}}$  are the parameters. If we want to estimate the ATET, we replace (14) with

$$\frac{1}{n} \sum_{i=1}^{n} \left\{ \left( \mathbf{x}_{i}' \widehat{\beta}_{11} + \widehat{\nu}_{i} \widehat{\beta}_{21} \right) \frac{n}{n_{t}} - \widehat{\text{POM0}} \frac{n}{n_{t}} - \widehat{\text{ATET}} \right\} = 0$$
 (15)

and if we want to estimate the potential-outcome means, we replace (14) with

$$\frac{1}{n} \sum_{i=1}^{n} \left\{ \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{11} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{21} \right) - \widehat{\text{POM1}} \right\} = 0 \tag{16}$$

where  $\widehat{\text{ATET}}$  and  $\widehat{\text{POM1}}$  are the parameters of the model, and  $n_t$  is the number of treated units.

For the exponential-mean outcome model, we replace  $\mathbf{x}_i'\widehat{\boldsymbol{\beta}}_{1j}+\widehat{\nu}_i\widehat{\boldsymbol{\beta}}_{2j}$  with  $\exp(\mathbf{x}_i'\widehat{\boldsymbol{\beta}}_{1j}+\widehat{\nu}_i\widehat{\boldsymbol{\beta}}_{2j})$  to obtain the residual equations in (10)–(16). For the probit outcome model, we replace (10) and (11) with the following:

$$\frac{1}{n} \sum_{i=1}^{n} t_{i} \mathbf{x}_{i}' \left\{ y_{i} \frac{\phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)}{\Phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)} - (1 - y_{i}) \frac{\phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)}{1 - \Phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)} \right\} = 0$$

$$\frac{1}{n} \sum_{i=1}^{n} (1 - t_{i}) \mathbf{x}_{i}' \left\{ y_{i} \frac{\phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)}{\Phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)} - (1 - y_{i}) \frac{\phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)}{1 - \Phi \left( \mathbf{x}_{i}' \widehat{\boldsymbol{\beta}}_{1j} + \widehat{\nu}_{i} \widehat{\boldsymbol{\beta}}_{2j} \right)} \right\} = 0$$

For the remaining equations,  $\mathbf{x}_i' \hat{\boldsymbol{\beta}}_{1j} + \hat{\nu}_i \hat{\boldsymbol{\beta}}_{2j}$  is replaced with  $\Phi(\mathbf{x}_i' \hat{\boldsymbol{\beta}}_{1j} + \hat{\nu}_i \hat{\boldsymbol{\beta}}_{2j})$ . The fractional probit model uses the same moment conditions as the probit model.

## Acknowledgment

We thank Jeffrey Wooldridge of Michigan State University for helpful conversations and advice about the literature and methods in this area.

## References

Cattaneo, M. D. 2010. Efficient semiparametric estimation of multi-valued treatment effects under ignorability. *Journal of Econometrics* 155: 138–154. https://doi.org/10.1016/j.jeconom.2009.09.023.

Cerulli, G. 2014. ivtreatreg: A command for fitting binary treatment models with heterogeneous response to treatment and unobservable selection. Stata Journal 14: 453–480.

—. 2022. Econometric Evaluation of Socio-Economic Programs: Theory and Applications. 2nd ed. Berlin: Springer.

Drukker, D. M. 2016. A generalized regression-adjustment estimator for average treatment effects from panel data. Stata Journal 16: 826–836.

Terza, J. V., A. Basu, and P. J. Rathouz. 2008. Two-stage residual inclusion estimation: Addressing endogeneity in health econometric modeling. *Journal of Health Economics* 27: 531–543. https://doi.org/10.1016/j.jhealeco.2007.09.009.

Wooldridge, J. M. 2010. Econometric Analysis of Cross Section and Panel Data. 2nd ed. Cambridge, MA: MIT Press.

— 2014. Quasi-maximum likelihood estimation and testing for nonlinear models with endogenous explanatory variables. Journal of Econometrics 182: 226–234. https://doi.org/10.1016/j.jeconom.2014.04.020.

#### Also see

- [CAUSAL] **eteffects postestimation** Postestimation tools for eteffects
- [CAUSAL] etregress Linear regression with endogenous treatment effects
- [CAUSAL] teffects Treatment-effects estimation for observational data
- [R] gmm Generalized method of moments estimation
- [R] **probit** Probit regression
- [R] **regress** Linear regression
- [U] 20 Estimation and postestimation commands

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright (c) 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.